



Programma HCC!Forth komende zaterdag

**Zaterdag 8 oktober 2016 op de bekende locatie:
het gebouw van peuterspeelzaal Juliaantje naast de Zuiderkapel aan de
Boslaan 1 in Bilthoven**

10:30 Zaal open en koffie voor vroege vogels

11:00 Voortgang van de werkgroepen

11:15 Een vervolg op het Egel-project, door Willem Ouwerkerk

13:00 Pauze

15:00 Sluiting.

Tot ziens



Zuiderkapel Boslaan 1 Bilthoven

Een vervolg op het Egel-project

Willem Ouwerkerk

Dit zijn de onderwerpen die behandeld worden:

- 1) Hexapod walking robot
- 2) Oled I2C graphic LCD 128x64 pixels
- 3) Reflection sensor TCRT5000 using ADC
- 4) L9110 PWM DC bridge driver
- 5) LiPo protection using onboard ADC
- 6) Lidar distance sensor
- 7) I2C Compass
- 8) I2C movement sensor
- 9) M(ini)ushi robot
- 10) Any other ideas



Oled GLCD

Small robot Pulse-width power control

(* Exxx - For noForth 2553 lp.0, C&V version: Small robot
Pulswidth power control with 4KHz PWM at P2.2 & P2.4
with a resolution of 1000 steps

P2.3 and P2.4 must be wired to one side of L9110 dual bridge
driver. P2.1 and P2.2 must be wired to the other side.

Take care for the maximum USB-driver current of 150mA!

Address 020 = P1IN, port-1 input register
Address 022 = P1DIR, port-1 direction register
Address 027 = P1REN, port-1 resistor enable
Address 029 = P2OUT, port-2 output with 8 leds
Address 02A = P2DIR, port-2 direction register
Address 02E = P2SEL, port-2 selection register
Address 180 = TA1CTL, timer a1 compare mode
Address 186 = TA1CCTL2, timer a1 output mode
Address 192 = TA1CCR0, timer a1 period timing
Address 196 = TA1CCR2, timer a1 Duty cycle

FEDCBA9876543210 bit-numbers
0000000000010000 - Choose output bit4 or bit5
0000000011000000 - toggle-set output
0000001000110100 - TA clear, up/down, SMCLK, no presc.

*)

hex

\ Period length is 1000 clock cycles (#CYCLUS)

dm 1000 constant #CYCLUS

value POWER

```
: >LEFT      ( 0 to #cyclus -- )
  #cyclus umin 196 !          \ TA1CCR2   Set pulselength left motor
  010 029 *bic ;              \ P2OUT

: >RIGHT      ( 0 to #cyclus -- )
  #cyclus umin 194 !          \ TA1CCR1   Set pulselength right motor
  002 029 *bic ;              \ P2OUT

: >PWM        ( +n -- )      dup >right >left ;          \ Pulselength for both
: FW          ( -- )         8 29 *bic power >pwm ;      \ Move forward primitive
: BACKW       ( -- )         8 29 *bis #cyclus 1+ power - >pwm ; \ Move backward
: STOP        ( -- )         8 29 *bic 0 >pwm ;          \ Stop moving
: FORW        ( -- )         backw dm 60 ms fw ;         \ Move forward
: LEFT        ( -- )         forw 1 >left ;              \ Stop left motor
: RIGHT       ( -- )         forw 1 >right ;             \ Stop right motor

\ Adjust speed in correct direction, but only when not stopped
: SPEED)      ( -- )
  196 @ if 8 29 bit* if backw exit then fw then ;

: SPEED       ( +n -- )      dm 800 umin dm 200 + to power speed) ; \ Set speed
: .SPEED      ( -- )         power dm 200 - . ;          \ Show current speed
: PWM-OFF     ( -- )         0 180 ! 014 02E *bic ; \ TA1CTL, P2SEL Deactivate PWM
```

```

\ PWM at P2.4 and stop motors
: SETUP-PWM ( -- )
  014 02E *bis 01E 02A *bis \ P2SEL, P2DIR Set PWM to output P2.2 & P2.4
  0 180 ! #cyclus 192 ! \ TA1CTL, TA1CCR0 Set period time
  040 186 ! \ TA1CCTL2 Set output mode positive pulse
  040 184 ! \ TA1CCTL1 Set output mode positive pulse (C0)
  234 180 ! \ TA1CTL Activate timer
  dm 200 speed stop ; \ Set basic speed, but stop motors

pwm-off shield MOTORS\

```

Reading of a reflection sensor type TCRT5000

```

\ Reading of a reflection sensor type TCRT5000
\ Using 3V3 to 3V6 and 220 Ohm in series with led.
\ 12 kOhm from collector to +3V3 to +3V6 the collector
\ is connected to an analog input from the MSP430
hex
: >LEDS ( b -- ) 029 c! ; \ P2OUT

\ ADC on and sample time at 64 clocks
: SETUP-ADC ( -- )
  02 1B0 *bic \ ADC10CTL0 Clear ENC
  80 04A c! \ ADC10AEO P1.7 = ADC in
  1810 1B0 ! ; \ ADC10CTL0 Sampletime 64 clocks, ADC on

\ We need to clear the ENC bit before setting a new input channel
: ADC ( +n -- u )
  02 1B0 *bic \ ADC10CTL0 Clear ENC
  F000 and 80 or 1B2 ! \ ADC10CTL1 Select input, MCLK/5
  03 1B0 *bis \ ADC10CTL0 Set ENC & ADC10SC
  begin 1 1B2 bit* 0= until \ ADC10CTL1 ADC10 busy?
  1B4 @ ; \ ADC10MEM Read result

: REFLECTION 7000 adc ; \ Read sensor level at P1.7

: SENS ( -- )
  setup-adc decimal \ Initialise
  begin
    reflection 0E / \ Read TCRT5000 and scale
    3F umin dup . >leds \ Print visual range on led array
  key? until 0 >leds ; \ Leds off

setup-adc shield SENSORS\ freeze

```

Ook iets te melden?

Stuur uw ideeën, programma's of projecten naar de redactie, zodat anderen daar ook kennis van kunnen nemen. Bijdragen liefst per E-mail, Uiterlijk 1 week voor de bijeenkomst, naar f.i.van.der.markt@kader.hcc.nl



HCC!Forth